

Escola Politécnica da Universidade de São Paulo  
**Departamento de Engenharia de Sistemas Eletrônicos - PSI**

**PSI-3452- Projeto de Circuitos Integrados Digitais e Analógicos**

**Lab 5: Projeto Eletrônico e Leiaute de um  
Amplificador Analógico- Prática (2022)**

## 1 – Objetivos

Desenvolver o projeto de um amplificador analógico simples. Entender o seu funcionamento e os passos que conduzem às dimensões de seus componentes. Realizar as simulações iniciais do projeto verificando a sua correção. Desenhar o leiaute deste amplificador com aplicação de técnicas específicas para circuitos analógicos, realizar a extração do circuito elétrico e simulação do mesmo. Comparar as previsões teóricas com as simulações pré e pós-leiaute.

## 3 – Parte Prática

### 3.1. Projeto do aluno baseado em equações simplificadas

O(A) aluno(a) gerará os parâmetros do seu projeto específico, na forma apresentado no exemplo da seção 2.4, mantendo as especificações de frequência e ganho.

Os parâmetros  $Wn1$  e  $Rcarga$  da eq. 6 deverão se ajustadas de acordo com o número USP, seguindo a atividade (provinha) proposta.

⇒ **Completar na folha de respostas com as informações solicitadas (item 3.1)**

### 3.2. Simulação do Projeto

Faça a simulação com o circuito completo (apenas o obtido teoricamente). Para isto, utilize como base o arquivo '**ampl\_ideal.sp**', dado no Moodle e mostrado a seguir. Edite-o em uma pasta referente a esta seção (~/**lab5/caso\_ideal**, por exemplo). O aluno deve alterar os números que estão em negrito, compatibilizando-os com os do seu projeto, obtidos através de seu número USP e de seus cálculos. Observe e analise a descrição do circuito para ter certeza que se trata do circuito desejado, além dos valores de polarização projetados.

**ATENÇÃO:** Tenha certeza que entende os comandos `.ac` e `.plot`

```
Simulacao no eldo de amplificador ideal
* nome lab5ideal.sp
.include /tools/mgc_tree/adk3_1/technology/ic/models/tsmc035.mod
M1 out pt1 0 0 n L=4e-07 W=100e-06
M2 pol pol 0 0 n L=4e-07 W=100e-06
Rdes pol pt1 8.7k
Cdes in pt1 1.82p
Rcarga fonte out 2.5k
Ipol pol 0 -0.66m
vin in 0 ac 0.1
vcc fonte 0 3.3
.ac dec 10 10e+4 10e+8
.plot ac vm(out) vm(in)
.end
```

Simule e obtenha a sua **curva de tensão de saída**. Deverá ser semelhante ao da Figura 10. Calcule o **ganho** ( $V_{out}/V_{in}$ ) na frequência planejada (use o  $V_{in}$  da sua descrição de circuito).




Figura 10 – resultado esperado de simulação

⇒ **Seguir a folha de respostas e completar com as informações solicitadas (item 3.2)**

### 3.3. Geração do Leiaute

Desenhe, usando o IC Station, o leiaute dos diversos componentes para, em seguida, conectá-los em forma de amplificador. Crie uma nova pasta (~/**lab5/caso\_leiaute**, por exemplo), Salve o seu leiaute com o nome **amplifier**, por exemplo. A seguir serão dadas as instruções para o desenho de cada um destes componentes.

#### 3.3.1. O par de transistores

Os seus transistores M1 e M2 devem seguir o esquema de leiaute da Figura 4 da Seção 2.2.2.

**a)** Vamos construir o M1 primeiro (a grade é em  $\lambda$  no IC Station):

Width (sub-bloco) =  $Wn1/(0.2)$  (use o seu valor)

Length (sub-bloco) =  $Ln/0.2 = 2\lambda$

**b)** Utilizando os comandos a seguir do IC Station, crie a base do transistor (de um sub-bloco):

- 1) Na palheta de projeto, selecione **DLA Device >> AddMos**
- 2) Insira na caixa de preenchimento à esquerda da tela, as dimensões do transistor.
- 3) Clique no ponto desejado na tela de leiaute. Você observará que trata-se de um transistor bastante estreito, com a razão de aspecto bem diferente de 1.

4) Deixe o transistor selecionado (F2 no vazio e depois F1 no bloco). No próprio AddMos, opte por **Edit > Fold Mos**.

5) Escolha 8 o número de *legs*, observe que isto resulta em um bloco próximo de um quadrado.

c) Crie uma linha de metal 1 horizontal (dimensões mínimas) acima do transistor para o dreno e conecte-a com as regiões "ímpares" de difusão, de forma intercalada, esticando as suas linhas de metal correspondentes, na forma descrita na Figura 4 e no leiaute exemplo da Figura 11.

d) Crie uma linha de metal 1 horizontal (dimensões mínimas) abaixo do transistor para a fonte e conecte-a com as regiões "pares" de difusão, de forma intercalada, esticando as suas linhas de metal correspondentes, na forma descrita na Figura 4 e no leiaute exemplo da Figura 11.

e) Inclua um contato de substrato p (lembre-se que trata-se de transistor tipo n).

1) O contato de substrato deve ser adicionado como foi feito ao se projetar o inversor em sessão de lab anterior, ou seja, utilizando o comando:

Add -> Instance

de '/tools/mgc\_tree/adk3\_1/technology/ic/process/tmsc035\_via/pwell\_contact'

f) Todas as portas devem ser interligadas de forma semelhante, porém usando-se uma linha horizontal de POLY1 (dimensões mínimas). A camada de POLY deve ter contato para metal 1 para futura interligação (ver Figura 11).

g) Duplique o transistor (para se ter M2); para isto, utilize o cursor e/ou as teclas f1 e f2 e selecione todo o transistor M1 já desenhado. Para duplicá-lo, utilize o comando 'Easy Edit / Copy' e selecione apenas mover horizontalmente. Deve resultar em uma figura parecida com a da Figura 11.

⇒ **Seguir a folha de respostas e completar com as informações solicitadas (item 3.3.1)**

### 3.3.2. Capacitor

Crie um capacitor com o valor por você projetado. Para isto, use o comando:

'DLA Device >> Cap >> Point Cap'

Na janela Object Editor selecione:

Capacitor Specification = Capacitance and Area Ratio

Capacitance = seu valor em fF obtido em **3.3.1**

Area Ratio = 1

Posicione o cursor onde você quer desenhar o capacitor clique o mouse no botão da esquerda.

**ATENÇÃO:** o capacitor apresenta **dois polos**. Descubra quais são eles no leiaute. Ambos os polos deverão se conectar a outros componentes por linhas de metal. Construa contatos se necessário.

### 3.3.3. Rcarga

Crie o resistor Rcarga usando o comando:

'DLA Device >> Res >> Point Res'

Na janela Object Editor selecione:

Type = HR

Resistor Specification = Resistance and width

Width = 5.0 (default)

Resistance = seu valor em Ohm obtido em 3.3

Number of Legs = [4]

Resistor Structure = series

Posicione o cursor onde você quer desenhar Rcarga e clique o botão esquerdo do mouse. Veja se o resultado é próximo de um quadrado. Senão, modifique o número de *legs*.

**ATENÇÃO:** o resistor apresenta **dois polos**. Descubra quais são eles no leiaute. Ambos os polos deverão se conectar a outros componentes por linhas de metal.

### 3.3.4. Rdes

Crie o resistor Rdes usando o comando:

'DLA Device >> Res >> Point Res'

Na janela Object Editor selecione:

Type = HR

Resistor Specification = Resistance and width

Width = 5.0 (default)

Resistance = seu valor em Ohm obtido em 3.3

Number of Legs = [9]

Resistor Structure = series

Posicione o cursor onde você quer desenhar Rdes e clique o botão esquerdo do mouse. Veja se o resultado é próximo de um quadrado. Senão, modifique o número de *legs*.

**ATENÇÃO:** o resistor apresenta **dois polos**. Descubra quais são eles no leiaute. Ambos os polos deverão se conectar a outros componentes por linhas de metal.

### 3.3.5. Juntando tudo

Após ter desenhado os diversos componentes e tê-los colocado de forma conveniente no leiaute, faça as ligações e coloque os rótulos dos portos (M1.port): Entrada, Saida, VCC e Ipol. O leiaute resultante deve ser parecido com o mostrado na Figura 11.

⇒ **Seguir a folha de respostas (item 3.3.5)**

### 3.4. Extração e Simulação

O aluno deve proceder com a extração do circuito **amplifier.sp**. Observe o arquivo extraído com cuidado, identificando os dois transistores, o capacitor e as resistências.

Faça a simulação no ELDO e verifique a conformidade dos resultados. É dado no Moodle o arquivo **ampl\_completo.sp** com a seguinte sintaxe, chamando o arquivo **amplifier.sp**:

```
Simulacao do leiaute o LAB5
.include /tools/mgc_tree/adk3_1/technology/ic/models/tsmc035.mod
.include './amplifier.sp'
X1 IN POL 0 VCC OUT LAB5
Ipol POL 0 -0.66m
vin IN 0 ac 0.1
vcc VCC 0 3.3
.ac dec 10 10e+4 10e+8
.plot ac vm(OUT) vm(IN)
.end
```

⇒ **Seguir a folha de respostas e completar com as informações solicitadas (item 3.4)**




Figura 11 – leiaute final

## Apendice A – Melhorias possíveis no circuito proposto.

### A.1 – Melhorando o espelho de corrente

Para reduzir o problema da impedância de saída do circuito, um espelho de corrente com melhor impedância de saída é proposto na figura A.1. Isto fará com que o ganho aumente, pois a impedância na saída será reduzida. O lado negativo em se ter transistores em série é que a tensão mínima de alimentação é aumentada.




Figura A.1 – Melhora no espelho de corrente

O arquivo que descreve este novo circuito é o mostrado a seguir:

Simulacao no eldo de amplificador ideal

```
.include /tools/mgc_tree/adk3_1/technology/ic/models/tsmc035.mod
M3 out pol pt2 0 n L=4e-07 W=100e-06
M4 pol pol pt3 0 n L=4e-07 W=100e-06
M1 pt2 pt1 0 0 n L=4e-07 W=100e-06
M2 pt3 pt3 0 0 n L=4e-07 W=100e-06
Rdes pt3 pt1 8.7k
Cdes in pt1 1.82p
Rcarga fonte out 2.5k
Ipol pol 0 -0.66m
vin in 0 ac 0.1
vcc fonte 0 3.3
.ac dec 10 10e+4 10e+8
.plot ac vm(out)
.end
```

Neste caso, a resposta é a mostrada na figura A.2 onde o ganho é da ordem de 15, ou seja, bem mais próximo do esperado.




Figura A.2 – Resposta em frequência

## A.2 – Usando um transistor PMOS como carga

Uma outra possível modificação é substituir o resistor  $R_c$  por um transistor canal P que esteja sempre conduzindo, conforme mostrado na figura A.3. Isto reduz a área, porém deve aumentar a distorção do sinal de saída.




Figura A.3 – Transistor canal P como carga

O arquivo usado nesta simulação é mostrado a seguir.

Simulação no eldo de amplificador ideal

```
.include /tools/mgc_tree/adk3_1/technology/ic/models/tsmc035.mod
M3 out pol pt2 0 n L=4e-07 W=100e-06
M4 pol pol pt3 0 n L=4e-07 W=100e-06
M1 pt2 pt1 0 0 n L=4e-07 W=100e-06
M2 pt3 pt3 0 0 n L=4e-07 W=100e-06
Rdes pt3 pt1 8.7k
Cdes in pt1 1.82p
M5 out 0 fonte fonte p W=54u L=10u
Ipol pol 0 -0.66m
vin in 0 ac 0.1
vcc fonte 0 3.3
.ac dec 10 10e+4 10e+8
.plot ac vm(out)
.end
```

O resultado da simulação é apresentado na figura A.4. Pode ser observado um aumento significativo do ganho. Pode-se esperar um aumento na distorção devido à não-linearidade do transistor canal P como carga.




Figura A.4 – Ganho usando transistor canal P como carga